

Math 2280-1
April 23, 2010

Revisiting the "guess the resonance game", which we first played using convolution formula, section 7.4, last Friday. We considered the undamped forced harmonic oscillator

$$x''(t) + x(t) = f(t)$$

with initial data $x(0)=v(0)=0$. We took the Laplace transform of this equation we deduce

$$X(s) = \frac{1 F(s)}{s^2 + 1}$$

so that the convolution theorem implies $x(t) = \sin * f(t)$. Since the unforced system has a natural angular frequency $\omega_0 = 1$, we expected resonance when the forcing function has the corresponding

period of $\frac{2\pi}{\omega_0} = 2\pi$. We discovered that our reasoning wasn't quite correct. Luckily, Fourier series lets

us answer the question of resonance completely, as discussed in today's class notes. In these Maple notes we illustrate that answer.

Example 0: We didn't do this when we considered convolution, but consider

$$f(t) = \cos(2t) - 2\cos(3t) .$$

It's (smallest) period is the least common multiple of π and $\frac{3\pi}{2}$. So, its period is the natural period of our oscillator equation, i.e. 2π . Nevertheless, we don't get resonance, since we can use superposition of non-resonating terms to get a particular solution to

$$x'(t) + x(t) = \cos(2t) - 2\cos(3t) .$$

> **restart:**

> **with(plots):**

> **with(DEtools):**

> **deqtn:=diff(x(t),t,t)+w0^2*x(t)=cos(w*t);**
dsolve(deqtn,x(t));


$$deqtn := \frac{d^2}{dt^2} x(t) + w_0^2 x(t) = \cos(wt)$$

$$x(t) = \sin(w_0 t) _C2 + \cos(w_0 t) _C1 - \frac{\cos(wt)}{-w_0^2 + w^2} \quad (1)$$

> **xp:=t->cos(2*t)/(-1+4) - 2*cos(3*t)/(-1+9);**

$$xp := t \rightarrow \frac{1}{3} \cos(2t) - \frac{1}{4} \cos(3t) \quad (2)$$

> **plot({cos(2*t)-2*cos(3*t),xp(t)},t=0..4*Pi,color=black,title='non-resonant forcing at the natural period');**




Example 1: A square wave forcing function with amplitude 1 and period 2π . This formula works until $t = 11\pi$.

```
> f:=t->-1+2*sum((-1)^n*Heaviside(t-n*Pi),n=-10..10);
#Heaviside was an early user of the unit step function
#and so Maple names it after him
```

$$f := t \rightarrow -1 + 2 \left(\sum_{n=-10}^{10} (-1)^n \text{Heaviside}(t - n\pi) \right)$$

(3)

```
> plot(f(t),t=-30..30,color=black,title='square wave forcing function');
```




```
> x:=t->int(sin(t-tau)*f(tau),tau=0..t);
#convolution formula for the solution
```

$$x := t \rightarrow \int_0^t \sin(t - \tau) f(\tau) d\tau$$

(4)

We got resonance:

```
> plot(x(t),t=0..30,color=black,title='resonance response?');
```



We can explain this with Fourier series. We've calculated Fourier series for the square wave, but we can also have Maple do this!

```
> Fouriercoeff:=proc(ff,L,a0,a,b) #ff=function, 2L=period
local m, #dummy letter to index coefficients
s; #domain variable
assume(m,integer);
a0:=simplify(1/L*int(ff(s),s=-L..L)):
a:=m->simplify(1/L*int(ff(s)*cos(Pi/L*m*s),s=-L..L)):
b:=m->simplify(1/L*int(ff(s)*sin(Pi/L*m*s),s=-L..L)):
end:
```

```

> squarewave:=t->-1+2*Heaviside(t):
Fouriercoeff(squarewave,Pi,Asq0,Asq,Bsq):
> Asq0;
Asq(n); #should be zero since squarewave is
#odd
Bsq(n); #should be zero when n is even,
#and *4/Pi)*1/n when n is odd.

```

$$0$$

$$0$$

$$-\frac{2(\cos(n\pi) - 1)}{n\pi}$$

(5)


So our square wave is approximated by

```

> squareapprox:=t-> 4/Pi*Sum(sin((2*k-1)*t)/(2*k-1),k=1..10);
plot(squareapprox(t),t=-10..10,color=black);

```

$$\text{squareapprox} := t \rightarrow \frac{4 \left(\sum_{k=1}^{10} \frac{\sin((2k-1)t)}{2k-1} \right)}{\pi}$$




And the piece of the particular solution corresponding to the first term $\frac{4 \sin(t)}{\pi}$ is $-\frac{2 t \cos(t)}{\pi}$, and is responsible for the resonance we saw earlier:

```

> plot({x(t),-2*t*cos(t)/Pi},t=0..30,color=black);

```




Example 2: tent function, same period.

> `tent:=t->int(f(u),u=0..t);`

$$tent := t \rightarrow \int_0^t f(u) \, du$$

(6)

> `plot(tent(t),t=-15..15,color=black, title='tent wave forcing function');`




> `y:=t->int(sin(t-tau)*tent(tau),tau=0..t);`

$$y := t \rightarrow \int_0^t \sin(t - \tau) \, tent(\tau) \, d\tau$$

(7)

> `plot(y(t),t=0..30,color=black, title='resonance response?');`



> `Fouriercoeff(tent,Pi,AT0,AT,BT):`

`assume(n,integer):`

`AT0;`

`AT(n);`


`BT(n);`

$$\frac{2 \left((-1)^{n-1} - 1 \right)}{\pi n^2}$$

(8)


> `tentapprox:=t->Pi/2+sum(AT(n)*cos(n*t),n=1..20):`

`plot(tentapprox(t),t=-10..10,color=black);`



And it's the first term $-\frac{4 \cos(t)}{\pi}$ and it's corresponding particular solution $-\frac{2 t \sin(t)}{\pi}$ that is causing resonance:

```
> plot({y(t), -2*t/Pi*sin(t)}, t=0..30, color=black);
```




Example 3: Now let's force with a period which is not the natural one. This square wave has period 2.

```
> h:=t->-1+2*sum((-1)^n*Heaviside(t-n), n=0..30);
```

$$h := t \rightarrow -1 + 2 \left(\sum_{n=0}^{30} (-1)^n \text{Heaviside}(t - n) \right)$$

(9)


```
> plot(h(t), t=0..30, color=black, title='forcing function');
```



```
> z:=t->int(sin(t-tau)*h(tau), tau=0..t);
plot(z(t), t=0..30, color=black, title='resonance response?');
```

$$z := t \mapsto \int_0^t \sin(t - \tau) h(\tau) \, d\tau$$

resonance response?




Example 4: A square wave which does not have the natural period, so we didn't expect resonance?

```
> k:=t->sum(Heaviside(t-4*Pi*n)-Heaviside(t-4*Pi*n-Pi),
n=-5..5);
```

$$k := t \rightarrow \sum_{n=-5}^5 (\text{Heaviside}(t - 4n\pi) - \text{Heaviside}(t - 4n\pi - \pi))$$

(10)

```
> plot(k(t), t=-30..30, color=black, title='forcing function, period = ?');
```




```
> w:=t->int(sin(t-tau)*k(tau), tau=0..t);
```

$$w := t \rightarrow \int_0^t \sin(t - \tau) k(\tau) d\tau$$

(11)

```
> plot(w(t), t=0..60, color=black, title='resonance response?');
```



Now we know what happened....our function has period 4π , but has $n=2$ non-zero Fourier coefficients:


```
> Fouriercoeff(k, 2*Pi, k0, ka, kb):
k0;
assume(n, integer);
ka(n);
kb(n);
```

$$\frac{1}{2}$$

$$\frac{\sin\left(\frac{1}{2}n\pi\right)}{\pi n} - \frac{-1 + \cos\left(\frac{1}{2}n\pi\right)}{\pi n}$$

(12)

```
> kapprox:=t->k0/2+sum(ka(n)*cos(1/2*n*t)+kb(n)*sin(1/2*n*t),n=1..20):
plot(kapprox(t),t=-20..20,color=black);
```



It was the $n=2$ sine term $\frac{\sin(t)}{\pi}$, which caused the resonance:

```
> plot({w(t),-t/(2*Pi)*cos(t)},t=0..30,color=black);
```

